

An Alternative Method for Resolution Improvement to Rotary Incremental Encoder

DORINA PURCARU, ELENA NICULESCU
 Faculty of Automation, Computers and Electronics
 University of Craiova
 13 Al. I. Cuza Street, 200585 Craiova
 ROMANIA

Abstract: - Many motions parameters can be measured and the motion sense detected based on the output signals of the rotary incremental encoder. The former method for resolution improvement supposes additional electronic circuits for detect and count the positive and negative fronts of the pulses generated by the sensor. This paper proposes an alternative method for resolution improvement and sense discrimination, adequate for the computerized study of rotary incremental encoders, using an experimental system that contains the sensor, an IBM-PC compatible computer and an intelligent interface with 80C552 microcontroller. Such experimental system is very useful in higher education or in research activities.

Key-Words: - rotary incremental encoder, measurement, displacement, resolution, sense discrimination.

1 Introduction

The rotary incremental encoder converts the angular displacement of its shaft into a digital electric signal. A collimated light beam is aimed against two radial reticules: static and moving (incremental disc). Light (that can pass through both reticules) drops on a photo sensor placed immediately over the disc. The disc is marked with a series of uniform lines in a single track around the perimeter. As the lines interrupt the light beam, "increments" of information are produced in the form of a square pulse train (output signal). The frequency of the pulses relates to the number of lines on the disc and the disc speed. The amplitude of the pulses relates to the excitation supply. The basic signal A provides information on single direction rotational movement. By using two scanning heads it is possible to produce a second wave train B with 90° displacement; the direction of rotation can be detected and the pulses are up or down counted for the position measurement.

The rotary incremental encoder can detect the motion sense and it is able to measure angular displacement and rotational speed of its disc when the sensor is connected to suitable electronic circuits and through proper mechanic link.

The angular displacement can be measured with different resolution. The former method for resolution improvement supposes additional electronic circuits for detect and count the positive and negative fronts of the pulses A and B. This paper proposes an alternative method for resolution

improvement, adequate for computerized study of rotary incremental encoder.

2 Former Method for Resolution Improvement

The signals A and B (depending on the angular displacement ϕ) are presented in Fig.1; their spatial period is $\Delta\phi$ (the spatial period of the incremental disc track) and we suppose the positive motion sense of the shaft. The signal B is phase-shifted with

4

and the addition of each pulse B_1 or B_2 is conditioned by the level of A . The associated electronic module generates two output signals, denoted P and M :

$$P = A_1 \cdot B + B_1 \cdot \bar{A} + A_2 \cdot \bar{B} + B_2 \cdot A, \quad (2)$$

$$M = A_1 \cdot \bar{B} + B_1 \cdot A + A_2 \cdot B + B_2 \cdot \bar{A}. \quad (3)$$




Fig. 1. The signal diagram for positive sense.

Three situations can appear:

- a) The motion sense is positive, when $M = 0$ and $P = A_1 + B_1 + A_2 + B_2$; (4)
- b) The motion sense is negative, when $P = 0$ and $M = A_1 + B_1 + A_2 + B_2$; (5)
- c) The incremental disc does not move, when $P = 0$ and $M = 0$.

The shaft position can be also measured if the pulses P are counted up and the pulses M are counted down in the position-counter [3].

3 An Alternative Method for Resolution Improvement

Measurement of plant outputs is essential for feedback control and is useful in performance evaluation of each process. Motion measurements are extremely useful in controlling mechanical responses and interactions in dynamic systems. There are many situations in which motion measurements are used for control purposes: in controlling tool machine operations, manipulator trajectory, different vehicles, and power-generating

plants, in fault detection or diagnostics etc. A one-to-one relationship may not always exist between a measuring device and a measured variable. Furthermore, the same measuring device may be used to measure different variables, using appropriate electronic circuits and data processing techniques [2,3]. So, an experimental system is very useful for supervise the function of a displacement sensor, in different working modes, and for test different methods for signal conditioning and data processing [4,5]. Such measurement system can be used in higher education or in research activities.

An experimental measurement system for studying the rotary incremental encoder is obtained using an IBM-PC compatible computer [7], an intelligent interface with 80C552 microcontroller [6], a signal conditioning subsystem and a mechanical subsystem. The main functions of the interface [5] are the motion control of the incremental disc and the local processing of some sensor data. The interface is connected to an IBM-PC compatible computer that assures a simple dialog with the user and data processing and interpretation; it also displays the measured variables and computes some sensor characteristics.

The microcontroller master program reads the digital inputs A and B , after each $115\mu s$. A routine assures the motion sense detection and the displacement measurement during this period.

Table 1. Logic levels of digital inputs A and B for positive motion sense

Logic levels before a new front		Logic levels after a new front	
a	b	a*	b*
0	0	1	0
1	0	1	1
1	1	0	1
0	1	0	0

Table 2. Logic levels of digital inputs A and B for negative motion sense

Logic levels before a new front		Logic levels after a new front	
a	b	a*	b*
0	0	0	1
0	1	1	1
1	1	1	0
1	0	0	0

The motion sense detection is based on the correlation between the logic levels of the signals A

and B, before and after each new detected front. This correlation is presented in Tables 1 and 2, for positive and negative, respectively motion sense.

The algorithm for motion sense detection and displacement measurement is presented below and in the flowchart from Fig. 2.

- a) Two variables (X and Y) memorise the last and the last but one, respectively logic levels of signals A and B.
- b) A logic operation XOR between these two variables assures the detection of each new front (positive or negative) of the pulses A or B; a non-zero result indicates the modification of the logic level of A or B (so, a new front).
- c) The last (denoted a and b) and the last but one (denoted a* and b*) logic levels of the digital inputs A and B form a number P that defines a row position in Table 3.




Fig. 2. Flowchart of the algorithm for the angular displacement measurement and motion sense discrimination.

```

$NOMOD51
NAME DETECT
$include (reg552.inc)
        DETECT_CODE      SEGMENT CODE
        DETECT_DATA      SEGMENT DATA
PUBLIC detecting, position, direction, old, act, vit, time, puls

RSEG DETECT_DATA
position: DS 4
    
```

Table 3. The correlation between the signal logic levels and the motion sense.

a b a* b*	P	S	a b a* b*	P	S
0 0 0 0	00	2	1 0 0 0	08	0
0 0 0 1	01	0	1 0 0 1	09	2
0 0 1 0	02	1	1 0 1 0	0A	2
0 0 1 1	03	2	1 0 1 1	0B	1
0 1 0 0	04	1	1 1 0 0	0C	2
0 1 0 1	05	2	1 1 0 1	0D	1
0 1 1 0	06	2	1 1 1 0	0E	0
0 1 1 1	07	0	1 1 1 1	0F	2

The last column of this table contains a sense code S:

- S=0 for negative motion sense,
- S=1 for positive motion sense,
- S=2 for uncertainty (unmodified levels or impossible situation).

So, Table 3 assures the motion sense detection.

- d) The position-counter counts up (for positive motion sense) or counts down (for negative motion sense) the last detected front of A or B. The displacement-counter always counts up the last detected front of A or B. The resulted

number is multiplied with $\frac{\Delta\phi}{4}$ for obtain the

angular position/displacement of the shaft.

The proposed method imposes drastic restrictions to the maximum rotational speed of the shaft. When the microcontroller reads the digital inputs A and B after each 115µs, all fronts of these signals are detected only if the rotational speed (measured in revolutions per minute) satisfies the condition

$$n[\text{r.p.m.}] \leq \frac{0.13}{N} \cdot 10^6, \tag{6}$$

where N is the number of steps (pulses A) per turn.

The source file of this application is presented below. The encoder under test is of SUMTAK origin, the LBL-007-1000 type; $\Delta\phi = 0.36\text{deg}$ for this encoder. Using the proposed method, the resolution of the displacement measurement is 0.09deg.

```

direction: DS 1
act: DS 1
old: DS 1
time: ds 2
puls: ds 2
vit: ds 2

RSEG DETECT_CODE

;table for motion sense detection
;sense code significance is the following:
;- 0 --- negative sense
;- 1 --- positive sense
;- 2 --- uncertainty
tabela_inc_dec:
db 2, 1, 0, 2
db 0, 2, 2, 1
db 1, 2, 2, 0
db 2, 0, 1, 2

;routine for sense discrimination and pulse counting
detecting:
    mov a,time+1 ;count the period
    dec time+1
    jnz no_timeHigh
    dec time+0 ;for speed measurement
no_timeHigh:
    mov a,time+1
    orl a,time+0
    jnz yet_noPeriod
;if the period is finished
;note the number of counted pulses (this number corresponds to rotative speed)
;copy the value of variable puls in variable vit
period_done:
    mov vit+0,puls+0
    mov vit+1,puls+1
    mov puls+0,#0
    mov puls+1,#0
    mov time+0,#021h ;counting restarts
    mov time+1,#0f6h ;for a period 500ms
yet_noPeriod:
    mov a, P4 ;read the levels of digital inputs (A and B)
    anl a, #03h
    mov act, a
    xrl a, old
    jnz avemUnFront ;a new front is detected if one bit is different
ret ;return in the main program if a new front is not detected
avemUnFront:
    mov a, old
    rl a
    rl a ;move previous levels in positions 2 3
    orl a, act ;and copy the values of A and B in positions 0 1
;resulted values is a row position in table for motion sense detection
    mov dptr, #tabela_inc_dec
    movc a,@a+dptr ;read the motion sense in table
    jz subtract ;decrement the position counter if the sense is negative
    jnb acc.1, addition ;value 2 in table signifies uncertainty
ret
addition:
    setb c ;note the positive (+) motion sense
    mov a,direction

```

```

    rlc    a            ;store the last 4 values of the sense
    anl    a,#0fh
    mov    direction, a    ;memorize the new history of the last 4 values
    mov    old,act        ;memorize the new levels of inputs A and B
    inc    puls+1        ;increment the pulse number
    mov    a,puls+1
    jnz    no_over256PulsesAdd
    inc    puls+0
no_over256PulsesAdd:
;increment the current position
    mov    a,position+03
    add    a,#01H
    mov    position+03,a
    mov    a,position+02
    addc   a,#0H
    mov    position+02,a
    mov    a,position+01
    addc   a,#0H
    mov    position+01,a
    mov    a,position
    addc   a,#0H
    mov    position,a
ret      ;return in the main program
substract:
    clr    c            ;note the negative (-) motion sense
    mov    a,direction
    rlc    a            ;store the last 4 values of the sense
    anl    a,#0fh
    mov    direction, a    ;memorize the new history of the last 4 values
    mov    old,act        ;memorize the new levels of inputs A and B
    inc    puls+1        ;increment the pulse number
    mov    a,puls+1
    jnz    no_over256PulsesSubb
    inc    puls+0
no_over256PulsesSubb:
;decrement the current position
    mov    a,position+03
    add    a,#0ffH
    mov    position+03,a
    mov    a,position+02
    addc   a,#0ffH
    mov    position+02,a
    mov    a,position+01
    addc   a,#0ffH
    mov    position+01,a
    mov    a,position
    addc   a,#0ffH
    mov    position,a
ret      ;return in the main program
END

```

4 Conclusion

The paper presents a soft method for resolution improvement, sense discrimination, displacement and speed measurement. This method is adequate for the computerized study of rotary incremental encoders, using an experimental system that contains the sensor, an IBM-PC compatible computer and an intelligent interface with 80C552 microcontroller. Such measurement system is very useful in higher

education or in research activities; it enables the test of different measurement methods or resolution improvement techniques and it displays the measured variables and other results. The proposed method for resolution improvement is not adequate for all applications of rotary incremental encoders.

References:

- [1]A. Helfrick, W. Cooper, *Modern Electronic Instrumentation and Measurement Techniques*, Prentice-Hall International Editions, 1990.
- [2]P. Koči, J. Tůma, Incremental Rotary Encoders Accuracy, *7th International Carpathian Control Conference*, Ostrava – Beskydy (Czech Republic), 2006, pp.257-260.
- [3]D.M. Purcaru, Computerized Study of the Linear Optical Incremental Encoder, *WSEAS Transactions on Circuits and Systems, Issue 2*, Vol. 2, April 2003, pp.351-355.
- [4]D.M. Purcaru, I. Purcaru, Experimental System for Studying Some Displacement Sensors, *10th International Symposium SINTES 10*, Craiova (Romania), 2000, Proceedings, pp.E106-E109.
- [5]I. Purcaru, D.M. Purcaru, Intelligent Interface for Studying Different Sensors, *10th International Symposium SINTES 10*, Craiova (Romania), 2000, Proceedings, pp.E75-E78.
- [6]J. Stewart, *The 8051 Microcontroller: Hardware, Software and Interfacing*, Prentice-Hall, 1993.
- [7]W.J. Tompkins, J.G. Webster, (eds), *Interfacing Sensors to the IBM-PC*, Englewood Cliffs, NJ: Prentice Hall, 1988.